USGS
science for a changing world

Techniques of Water-Resources Investigations
of the United States Geological Survey

Chapter A1

# A MODULAR THREE-DIMENSIONAL FINITE-DIFFERENCE GROUND-WATER FLOW MODEL

By Michael G. McDonald and
Arlen W. Harbaugh

This chapter supersedes U.S. Geological Survey Open-File Report 83–875

Book 6

MODELING TECHNIQUES

## Narrative for Module SIP1AP

Module SIP1AP performs one iteration of the Strongly Implicit Procedure (SIP) algorithm for solving the flow equation. To save computational time, all arrays are declared one dimensional. The one-dimensional indexes are calculated from the layer, row, and column indexes normally used to access the arrays in three dimensions. Improvement in computational time is achieved because knowledge of the geometry is used to increase computational efficiency, and because calculations are not repeated for identical indices as would be done by internal FORTRAN addressing routines if three-dimensional subscripts were used.

This module is somewhat complex, partly because the SIP solution process requires that the same calculations be performed with two methods of ordering the equations. This is implemented by a generalized algorithm that uses the same computer statements to handle both ordering schemes. Checks are made to detect which ordering scheme is used, and array indexes are calculated accordingly.

Double precision is used for most calculations in this module in order to allow accurate answers to be calculated for a wide range of problems. Mixed precision arithmetic has been avoided by setting double-precision variables equal to single-precision values and then using the double-precision variables to generate fully double-precision expressions, and where necessary by doing the reverse to generate fully single precision expressions.

In the explanations below, no attempt has been made to discuss each use of an assignment statement to change precision, because of the large amount of text this would require. However, when changing this module, care should be used to maintain expressions that have unmixed precision, as mixed precision expressions can cause erroneous results with some compilers.

Module SIP1AP performs its functions in the following order:

1. If the user has specified (IPCALC ≠ 0) that iteration parameters should be calculated by the program, CALL submodule SSIP1I to calculate both the seed and the parameters.

2. Assign values to fields that are constant during an iteration.

3. Initialize the variables that track maximum head change during an iteration.

4. Clear SIP work arrays.

5. Determine the ordering of equations and set the ordering flag (IDIR) accordingly. This flag alternates between 1 and -1 each iteration. Calculate indexes IDNRC and IDNCOL which are used when calculating locations of neighboring cells.

6. Calculate the matrix [U] and intermediate vector {V} using forward substitution. The elements in matrix [L] are used as they are calculated; therefore, they are not saved. In the explanation of SIP concepts, the diagonals in the matrix [U] were designated "e," "f," and "g." The corresponding field names in the program are EL (e lower case), FL (f lower case),

and GL (g lower case). Similarly, the diagonals in the [L] array which are "a," "b," "c," and "d" in the explanation are "AL," "BL," "CL," and "DL" in the program. The codes for the diagonals in matrix [A] in the explanation are the same in the program. The codes for diagonals in [A+B] in the explanation are followed by a "P" in the program. Hence, Z' in the explanation is ZP (Z prime) in the program. The intermediate vector {V} in the explanation is the array "V" in the program.

(a) Set current cell indexes, II, JJ, KK. For normal ordering, the equation order is the same as the order of the loop indexes I,J,K. For reverse ordering, loop indexes I and K are inverted to produce the proper sequence of cells.

(b) Calculate the one-dimensional subscript of the current cell. If this cell is constant head or no flow, skip calculations for this cell and go on to the next.

(c) Calculate the one-dimensional subscripts for the six neighboring cells.

(d) Calculate the one-dimensional subscripts for conductance to each of the six neighboring cells. Since conductances between cells are assigned to array elements at specific cells (for example, CR(I,J,K) stores conductance between cells I,J,K and I,J+1,K), the four or five conductance subscripts are not simply the cell locations of the six neighboring cells as calculated in step 6(c). Also, the subscripts depend on equation ordering.

(e) Calculate or assign variables that are required for forward substitution and involve neighboring cells. Whenever a neighboring cell is outside of the grid, the variables are set to zero.

(1) Neighboring cell is one row back.

(2) Neighboring cell is one row ahead.

(3) Neighboring cell is one column back.

(4) Neighboring cell is one column ahead.

(5) Neighboring cell is one layer back.

(6) Neighboring cell is one layer ahead.

(f) Calculate the components of the upper and lower triangular matrices [U] and [L], which are the factors of matrix [A+B].

(g) Calculate the residual {RES}. The calculation of HNW times HCOF is done in single precision so that the calculation will have precision comparable to similar calculations made in the formulation modules, all of which use single precision.

(h) Calculate the intermediate vector {V}, which is stored in array V. This step completes the forward-substitution process for one cell.

7. Step through the cells solving for head change using back substitution.

(a) Set current cell indexes II, JJ, KK. The ordering is the reverse of that used for forward substitution (step 6(a)).

(b) Calculate the one-dimensional subscript of the current cell. If this cell is constant head or no flow, skip calculations for this cell and go to the next.

(c) Calculate the one-dimensional subscripts for the three neighboring cells behind (relative to the direction of the back-substitution ordering) the current cell.

(d) Back substitute, solving for head change. Store head change in array V in place of the intermediate values of vector {V}. This doubling up of storage is used to save the cost of additional computer storage.

(e) Save the value of head change whose absolute value is largest during this iteration. Also, save the cell location where this head change occurred and the absolute value of the head change.

(f) Add the head change this iteration to head from the previous iteration to get a new estimate of head.

8. Store the head change whose absolute value is greatest this iteration and its cell location in arrays HDCG and LRCH. These may be printed in step 10 at the end of the time step. Set the convergence flag to one if the convergence criterion is met.

9. If the iteration is complete, print the number of iterations for the step; otherwise, RETURN.

10. Print the maximum head change and cell location each iteration if the SIP printout interval (IPRSIP) is reached. Printout occurs at the end of a stress period regardless of the interval.

11. RETURN.

## Flow Chart for Module SIP1AP

IPCALC is a flag. If it is set equal to one, the program calculates a seed from which iteration parameters are calculated. It may be set by the user at the beginning of the simulation. It is cleared during the first iterations. SSIP1I will never be called more than once. If IPCALC is not set equal to zero, the user specifies the seed for the iteration parameters.

IDIR indicates whether the ordering of equations is normal (1) or reverse (-1).

SSIP1I is a submodule which calculates iteration parameters.



Single Cell Index: In this module, a single index is used to identify each cell. This is in opposition to the three indices (I,J,K) used in most other modules.



```
      SUBROUTINE SIP1AP(HNEW,IBOUND,CR,CC,CV,HCOF,RHS,EL,FL,GL,V,
     1      W,HDCG,LRCH,NPARM,KITER,HCLOSE,ACCL,ICNVG,KSTP,KPER,
     2      IPCALC,IPRSIP,MXITER,NSTP,NCOL,NROW,NLAY,NODES,IOUT)
C-----VERSION 1656 24JUL1987 SIP1AP
C
C     ******************************************************************
C     SOLUTION BY THE STRONGLY IMPLICIT PROCEDURE -- 1 ITERATION
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      DOUBLE PRECISION HNEW,DITPAR,AC,HHCOF,RRHS,XI,DZERO,DONE,RES
      DOUBLE PRECISION Z,B,D,E,F,H,S,AP,TP,CP,GP,UP,RP
      DOUBLE PRECISION ZHNEW,BHNEW,DHNEW,FHNEW,HHNEW,SHNEW
      DOUBLE PRECISION AL,BL,CL,DL,ELNCL,FLNCL,GLNCL
      DOUBLE PRECISION ELNRL,FLNRL,GLNRL,ELNLL,FLNLL,GLNLL
      DOUBLE PRECISION VNRL,VNCL,VNLL,ELXI,FLXI,GLXI,VN,HCFHNW
C
      DIMENSION HNEW(NODES), IBOUND(NODES), CR(NODES), CC(NODES),
     1  CV(NODES), HCOF(NODES), RHS(NODES), EL(NODES), FL(NODES),
     2  GL(NODES), V(NODES), W(NPARM), HDCG(MXITER), LRCH(3,MXITER)
C     ------------------------------------------------------------------
C
C1------CALCULATE ITERATION PARAMETERS IF FLAG IS SET.  THEN
C1------CLEAR THE FLAG SO THAT CALCULATION IS DONE ONLY ONCE.
      IF(IPCALC.NE.0)
     1     CALL SSIP1I(CR,CC,CV,IBOUND,NPARM,W,NCOL,NROW,NLAY,IOUT)
      IPCALC=0
C
C2------ASSIGN VALUES TO FIELDS THAT ARE CONSTANT DURING AN ITERATION
      DZERO=0.
      DONE=1.
      AC=ACCL
      NRC=NROW*NCOL
      NTH=MOD(KITER-1,NPARM)+1
      DITPAR=W(NTH)
C
C3------INITIALIZE VARIABLE THAT TRACKS MAXIMUM HEAD CHANGE DURING
C3------THE ITERATION
      BIGG=0.
C
C4------CLEAR SIP WORK ARRAYS.
      DO 100 I=1,NODES
      EL(I)=0.
      FL(I)=0.
      GL(I)=0.
  100 V(I)=0.
C
C5------SET NORMAL/REVERSE EQUATION ORDERING FLAG (1 OR -1) AND
C5------CALCULATE INDEXES DEPENDENT ON ORDERING
      IDIR=1
      IF(MOD(KITER,2).EQ.0)IDIR=-1
      IDNRC=IDIR*NRC
      IDNCOL=IDIR*NCOL
C
C6------STEP THROUGH CELLS CALCULATING INTERMEDIATE VECTOR V
C6------USING FORWARD SUBSTITUTION
      DO 150 K=1,NLAY
      DO 150 I=1,NROW
      DO 150 J=1,NCOL
C
C6A-----SET UP CURRENT CELL LOCATION INDEXES.  THESE ARE DEPENDENT
C6A-----ON THE DIRECTION OF EQUATION ORDERING.
      IF(IDIR.LE.0)GO TO 120
      II=I
      JJ=J
      KK=K
      GO TO 122
  120 II=NROW-I+1
      JJ=J
      KK=NLAY-K+1
```

```
C
C6B-----CALCULATE 1 DIMENSIONAL SUBSCRIPT OF CURRENT CELL AND
C6B-----SKIP CALCULATIONS IF CELL IS NOFLOW OR CONSTANT HEAD
  122 N=JJ+(II-1)*NCOL+(KK-1)*NRC
      IF(IBOUND(N).LE.0)GO TO 150
C
C6C-----CALCULATE 1 DIMENSIONAL SUBSCRIPTS FOR LOCATING THE 6
C6C-----SURROUNDING CELLS
      NRN=N+IDNCOL
      NRL=N-IDNCOL
      NCN=N+1
      NCL=N-1
      NLN=N+IDNRC
      NLL=N-IDNRC
C
C6D-----CALCULATE 1 DIMENSIONAL SUBSCRIPTS FOR CONDUCTANCE TO THE 6
C6D-----SURROUNDING CELLS.  THESE DEPEND ON ORDERING OF EQUATIONS.
      IF(IDIR.LE.0)GO TO 124
      NCF=N
      NCD=NCL
      NRB=NRL
      NRH=N
      NLS=N
      NLZ=NLL
      GO TO 126
  124 NCF=N
      NCD=NCL
      NRB=N
      NRH=NRN
      NLS=NLN
      NLZ=N
C
C6E-----ASSIGN VARIABLES IN MATRICES A & U INVOLVING ADJACENT CELLS
C6E1----NEIGHBOR IS 1 ROW BACK
  126 B=DZERO
      ELNRL=DZERO
      FLNRL=DZERO
      GLNRL=DZERO
      BHNEW=DZERO
      VNRL=DZERO
      IF(I.EQ.1) GO TO 128
      B=CC(NRB)
      ELNRL=EL(NRL)
      FLNRL=FL(NRL)
      GLNRL=GL(NRL)
      BHNEW=B*HNEW(NRL)
      VNRL=V(NRL)
C
C6E2----NEIGHBOR IS 1 ROW AHEAD
  128 H=DZERO
      HHNEW=DZERO
      IF(I.EQ.NROW) GO TO 130
      H=CC(NRH)
      HHNEW=H*HNEW(NRN)
C
C6E3----NEIGHBOR IS 1 COLUMN BACK
  130 D=DZERO
      ELNCL=DZERO
      FLNCL=DZERO
      GLNCL=DZERO
      DHNEW=DZERO
      VNCL=DZERO
      IF(J.EQ.1) GO TO 132
      D=CR(NCD)
      ELNCL=EL(NCL)
      FLNCL=FL(NCL)
      GLNCL=GL(NCL)
      DHNEW=D*HNEW(NCL)
      VNCL=V(NCL)
```

```
C
C6E4----NEIGHBOR IS 1 COLUMN AHEAD
  132 F=DZERO
      FHNEW=DZERO
      IF(J.EQ.NCOL) GO TO 134
      F=CR(NCF)
      FHNEW=F*HNEW(NCN)
C
C6E5----NEIGHBOR IS 1 LAYER BEHIND
  134 Z=DZERO
      ELNLL=DZERO
      FLNLL=DZERO
      GLNLL=DZERO
      ZHNEW=DZERO
      VNLL=DZERO
      IF(K.EQ.1) GO TO 136
      Z=CV(NLZ)
      ELNLL=EL(NLL)
      FLNLL=FL(NLL)
      GLNLL=GL(NLL)
      ZHNEW=Z*HNEW(NLL)
      VNLL=V(NLL)
C
C6E6----NEIGHBOR IS 1 LAYER AHEAD
  136 S=DZERO
      SHNEW=DZERO
      IF(K.EQ.NLAY) GO TO 138
      S=CV(NLS)
      SHNEW=S*HNEW(NLN)
C
C6E7----CALCULATE THE NEGATIVE SUM OF ALL CONDUCTANCES TO NEIGHBORING
C6E7----CELLS
  138 E=-Z-B-D-F-H-S
C
C6F-----CALCULATE COMPONENTS OF THE UPPER AND LOWER MATRICES, WHICH
C6F-----ARE THE FACTORS OF MATRIX (A+B)
      AL=Z/(DONE+DITPAR*(ELNLL+FLNLL))
      BL=B/(DONE+DITPAR*(ELNRL+GLNRL))
      CL=D/(DONE+DITPAR*(FLNCL+GLNCL))
      AP=AL*ELNLL
      CP=BL*ELNRL
      GP=CL*FLNCL
      RP=CL*GLNCL
      TP=AL*FLNLL
      UP=BL*GLNRL
      HHCOF=HCOF(N)
      DL=E+HHCOF+DITPAR*(AP+TP+CP+GP+UP+RP)-AL*GLNLL-BL*FLNRL-CL*ELNCL
      EL(N)=(F-DITPAR*(AP+CP))/DL
      FL(N)=(H-DITPAR*(TP+GP))/DL
      GL(N)=(S-DITPAR*(RP+UP))/DL
C
C6G-----CALCULATE THE RESIDUAL
      RRHS=RHS(N)
      HNW=HNEW(N)
      HCFHNW=HNW*HCOF(N)
      RES=RRHS-ZHNEW-BHNEW-DHNEW-E*HNEW(N)-HCFHNW-FHNEW-HHNEW-SHNEW
C
C6H-----CALCULATE THE INTERMEDIATE VECTOR V
      V(N)=(AC*RES-AL*VNLL-BL*VNRL-CL*VNCL)/DL
C
  150 CONTINUE
C
C7------STEP THROUGH EACH CELL AND SOLVE FOR HEAD CHANGE BY BACK
C7------SUBSTITUTION
      DO 160 K=1,NLAY
      DO 160 I=1,NROW
      DO 160 J=1,NCOL
C
```

```
C7A-----SET UP CURRENT CELL LOCATION INDEXES.  THESE ARE DEPENDENT
C7A-----ON THE DIRECTION OF EQUATION ORDERING.
      IF(IDIR.LT.0) GO TO 152
      KK=NLAY-K+1
      II=NROW-I+1
      JJ=NCOL-J+1
      GO TO 154
  152 KK=K
      II=I
      JJ=NCOL-J+1
C
C7B-----CALCULATE 1 DIMENSIONAL SUBSCRIPT OF CURRENT CELL AND
C7B-----SKIP CALCULATIONS IF CELL IS NOFLOW OR CONSTANT HEAD
  154 N=JJ+(II-1)*NCOL+(KK-1)*NRC
      IF(IBOUND(N).LE.0)GO TO 160
C
C7C-----CALCULATE 1 DIMENSIONAL SUBSCRIPTS FOR THE 3 NEIGHBORING CELLS
C7C-----BEHIND (RELATIVE TO THE DIRECTION OF THE BACK SUBSTITUTION
C7C-----ORDERING) THE CURRRENT CELL.
      NC=N+1
      NR=N+IDNCOL
      NL=N+IDNRC
C
C7D-----BACK SUBSTITUTE, STORING HEAD CHANGE IN ARRAY V IN PLACE OF
C7D-----INTERMEDIATE FORWARD SUBSTITUTION VALUES.
      ELXI=DZERO
      FLXI=DZERO
      GLXI=DZERO
      IF(JJ.NE.NCOL) ELXI=EL(N)*V(NC)
      IF(I.NE.1) FLXI=FL(N)*V(NR)
      IF(K.NE.1) GLXI=GL(N)*V(NL)
      VN=V(N)
      V(N)=VN-ELXI-FLXI-GLXI
C
C7E-----GET THE ABSOLUTE HEAD CHANGE. IF IT IS MAX OVER GRID SO FAR.
C7E-----THEN SAVE IT ALONG WITH CELL INDICES AND HEAD CHANGE.
      TCHK=ABS(V(N))
      IF (TCHK.LE.BIGG) GO TO 155
      BIGG=TCHK
      BIG=V(N)
      IB=II
      JB=JJ
      KB=KK
C
C7F-----ADD HEAD CHANGE THIS ITERATION TO HEAD FROM THE PREVIOUS
C7F-----ITERATION TO GET A NEW ESTIMATE OF HEAD.
  155 XI=V(N)
      HNEW(N)=HNEW(N)+XI
C
  160 CONTINUE
C
C8------STORE THE LARGEST ABSOLUTE HEAD CHANGE (THIS ITERATION) AND
C8------AND ITS LOCATION.
      HDCG(KITER)=BIG
      LRCH(1,KITER)=KB
      LRCH(2,KITER)=IB
      LRCH(3,KITER)=JB
      ICNVG=0
      IF(BIGG.LE.HCLOSE) ICNVG=1
C
C9------IF END OF TIME STEP, PRINT # OF ITERATIONS THIS STEP
      IF(ICNVG.EQ.0 .AND. KITER.NE.MXITER) GO TO 600
      IF(KSTP.EQ.1) WRITE(IOUT,500)
  500 FORMAT(1HO)
      WRITE(IOUT,501) KITER,KSTP,KPER
  501 FORMAT(1X,I5,' ITERATIONS FOR TIME STEP',I4,' IN STRESS PERIOD',
     1        I3)
C
C10-----PRINT HEAD CHANGE EACH ITERATION IF PRINTOUT INTERVAL IS REACHED
      IF(ICNVG.EQ.0 .OR. KSTP.EQ.NSTP .OR. MOD(KSTP,IPRSIP).EQ.0)
     1      CALL SSIP1P(HDCG,LRCH,KITER,MXITER,IOUT)
C
C11-----RETURN
600   RETURN
C
      END
```

## List of Variables for Module SIP1AP

| Variable | Range | Definition |
|---|---|---|
| AC | Module | Double-precision acceleration parameter (ACCL). |
| ACCL | Package | Acceleration parameter. |
| AL | Module | Diagonal from the lower factor. (AL stands for A-lower case.) |
| AP | Module | Diagonal element in the modified coefficient matrix. (AP stands for A-prime.) |
| B | Module | Diagonal label in the coefficient matrix--conductance from the adjacent node which is in the last row. |
| BHNEW | Module | Head in the adjacent cell which is in the last row. |
| BIG | Module | Largest head change for an iteration. |
| BIGG | Module | Largest absolute value of head change for an iteration. |
| BL | Module | Diagonal from the lower factor. (BL stands for B-lower case.) |
| CC | Global | DIMENSION (NCOL,NROW,NLAY), Conductance in the column direction. CC(J,I,K) contains conductance between nodes (J,I,K) and (J,I+1,K). |
| CL | Module | Diagonal from the lower factor. (CL stands for C-lower case.) |
| CP | Module | Diagonal element in the modified coefficient matrix. (CP stands for C-prime.) |
| CR | Global | DIMENSION (NCOL,NROW,NLAY), Conductance in the row direction. CR(J,I,K) contains conductance between nodes (J,I,K) and (J+1,I,K) |
| CV | Global | DIMENSION (NCOL,NROW,NLAY-1), Conductance in the vertical direction. CV(J,I,K) contains conductance between nodes (J,I,K) and (J,I,K+1). |
| D | Module | Diagonal label in the coefficient matrix. Conductance from the adjacent node which is in the last column. |
| DHNEW | Module | Head in the adjacent cell which is in the last column. |
| DITPAR | Module | Double-precision iteration parameter. |
| DL | Module | Diagonal from the lower factor. (DL stands for D-lower case.) |
| DONE | Module | Double-precision field containing a one. |
| DZERO | Module | Double-precision field containing a zero. |
| E | Module | Main diagonal in the coefficient matrix. |
| EL | Module | DIMENSION (NODES), Diagonal from the upper factor. (EL stands for E-lower case.) |
| ELNCL | Module | EL (E-lower case) from the cell in the last column. |
| ELNLL | Module | EL (E-lower case) from the cell in the last layer. |
| ELNRL | Module | EL (E-lower case) from the cell in the last row. |
| ELXI | Module | Intermediate result. |
| F | Module | Diagonal label in the coefficient matrix--conductance from the adjacent node which is in the next column. |
| FHNEW | Module | Head in the adjacent cell which is in the next column. |
| FL | Module | DIMENSION (NODES), Diagonal from the upper factor. (FL stands for F-lower case.) |
| FLNCL | Module | FL (F-lower case) from the cell in the last column. |
| FLNLL | Module | FL (F-lower case) from the cell in the last layer. |
| FLNRL | Module | FL (F-lower case) from the cell in the last row. |

List of Variables for Module SIP1AP (Continued)

| Variable | Range | Definition |
|---|---|---|
| FLXI | Module | Intermediate result. |
| GL | Module | DIMENSION (NODES), Diagonal from the upper factor. (GL stands for G-lower case.) |
| GLNCL | Module | GL (G-lower case) from the cell in the last column. |
| GLNLL | Module | GL (G-lower case) from the cell in the last layer. |
| GLNRL | Module | GL (G-lower case) from the cell in the last row. |
| GLXI | Module | Intermediate result. |
| GP | Module | Diagonal element in the modified coefficient matrix. (GP stands for G-prime.) |
| H | Module | Diagonal label in the coefficient matrix. Conductance from the adjacent node which is in the next row. |
| HCFHNW | Module | Product of head and HCOF for a cell. |
| HCLOSE | Package | Closure criterion for the iterative procedure. |
| HCOF | Global | DIMENSION (NCOL,NROW,NLAY), Coefficient of head in the cell (J,I,K) in the finite-difference equation. |
| HDCG | Package | DIMENSION (MXITER), Maximum head change for each iteration. |
| HHCOF | Module | Double-precision HCOF. |
| HHNEW | Module | Head in the adjacent cell which is in the next row. |
| HNEW | Global | DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration. |
| HNW | Module | Temporary field for HNEW(N). |
| I | Module | Index for nodes and rows. |
| IB | Module | Row number of the cell having the largest head change. |
| IBOUND | Global | DIMENSION (NCOL,NROW,NLAY), Status of each cell.<br>< 0, constant-head cell<br>= 0, inactive cell<br>> 0, variable-head cell |
| ICNVG | Global | Flag is set equal to one when the iteration procedure has converged. |
| IDIR | Module | Indicator for direction of solution algorithm.<br>+1 - forward<br>-1 - reverse |
| IDNCOL | Module | Intermediate result used to calculate indices. |
| IDNRC | Module | Intermediate result used to calculate indices. |
| II | Module | Row number. |
| IOUT | Global | Primary unit number for all printed output. IOUT = 6. |
| IPCALC | Package | Flag.<br>= 0, iteration parameter seed (WSEED) is entered by the user.<br>= 1, seed is calculated in the program. |
| IPRSIP | Package | Frequency (in time steps) with which the maximum head changes for each iteration will be printed. |
| J | Module | Index for columns. |
| JB | Module | Column number of the cell having the largest head change. |
| JJ | Module | Column index. |
| K | Module | Index for layers. |
| KB | Module | Layer of the cell having the largest head change. |
| KITER | Global | Iteration counter. Reset at the start of each time step. |

List of Variables for Module SIP1AP (Continued)

| Variable | Range | Definition |
|---|---|---|
| KK | Module | Layer index. |
| KPER | Global | Stress period counter. |
| KSTP | Global | Time step counter. Reset at the start of each stress period. |
| LRCH | Package | DIMENSION (MXITER), Layer, row, and column of the cell containing the maximum head change (HDCG) for each iteration. |
| MXITER | Package | Maximum number of iterations. |
| N | Module | Cell index. |
| NC | Module | Index for the adjacent cell in the last column. |
| NCD | Module | One-dimensional subscript of conductance to the adjacent cell which is in the last column. |
| NCF | Module | One-dimensional subscript of conductance to the adjacent cell which is in the next column. |
| NCL | Module | One-dimensional subscript of the cell index of the adjacent cell which is in the last column. |
| NCN | Module | One-dimensional subscript of the cell index of the adjacent cell which is in the next column. |
| NCOL | Global | Number of columns in the grid. |
| NL | Module | Index for the adjacent cell in the last layer. |
| NLAY | Global | Number of layers in the grid. |
| NLL | Module | One-dimensional subscript of the cell index of the adjacent cell which is in the last layer. |
| NLN | Module | One-dimensional subscript of the cell index of the adjacent cell which is in the next layer. |
| NLS | Module | One-dimensional subscript of conductance to the adjacent cell which is in the next layer. |
| NLZ | Module | One-dimensional subscript of conductance to the adjacent cell which is in the last layer. |
| NODES | Global | Number of cells (nodes) in the finite-difference grid. |
| NPARM | Package | Number of iteration parameters. |
| NR | Module | Index for the adjacent cell in the last row. |
| NRB | Module | One-dimensional subscript of conductance to the adjacent cell which is in the last row. |
| NRC | Module | Number of cells in the layer. |
| NRH | Module | One-dimensional subscript of conductance to the adjacent cell which is in the next row. |
| NRL | Module | One-dimensional subscript of the cell index of the adjacent cell which is in the last row. |
| NRN | Module | One-dimensional subscript of the cell index of the adjacent cell which is in the next row. |
| NROW | Global | Number of rows in the grid. |
| NSTP | Global | Number of time steps in the current stress period. |
| NTH | Module | Index for iteration parameters. |
| RES | Module | Residual. |
| RHS | Global | DIMENSION (NCOL,NROW,NLAY), Right hand side of the finite-difference equation. RHS is an accumulation of terms from several different packages. |

## List of Variables for Module SIP1AP (Continued)

| Variable | Range | Definition |
|---|---|---|
| RP | Module | Diagonal element in the modified coefficient matrix. (RP stands for R-prime.) |
| RRHS | Module | Double-precision right hand side of the equation. |
| S | Module | Diagonal label in the coefficient matrix--conductance from the adjacent node which is in the next layer. |
| SHNEW | Module | Head in the adjacent cell which is in the next layer. |
| TCHK | Module | Absolute value of head change for a single cell. |
| TP | Module | Diagonal element in the modified coefficient matrix. (TP stands for T-prime.) |
| UP | Module | Diagonal element in the modified coefficient matrix. (UP stands for U-prime.) |
| V | Package | DIMENSION (NODES), Intermediate result. |
| VN | Module | Temporary double-precision V(N). |
| VNCL | Module | Element in the intermediate vector for the cell in the last column. |
| VNLL | Module | Element in the intermediate vector for the cell in the last layer. |
| VNRL | Module | Element in the intermediate vector for the cell in the last row. |
| W | Package | DIMENSION (NPARM), Iteration parameters. |
| XI | Module | Double-precision V(N). |
| Z | Module | Diagonal label in the coefficient matrix--conductance from the adjacent node which is in the last layer. |
| ZHNEW | Module | Head in the adjacent cell which is in the last layer. |

## Narrative for Module SSIP1P

Submodule SSIP1P prints the largest value of head change (HDCG) out of all cells for each iteration of a time step. Also printed is the cell location (LRCH) where the change occurs. The submodule is so short that no numbered comments are used and no flow chart is provided.

```
      SUBROUTINE SSIP1P(HDCG,LRCH,KITER,MXITER,IOUT)
C
C
C-----VERSION 1636 24JUL1987 SSIP1P
C     ******************************************************************
C     PRINT MAXIMUM HEAD CHANGE FOR EACH ITERATION DURING A TIME STEP
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
C
      DIMENSION HDCG(MXITER), LRCH(3,MXITER)
C     ------------------------------------------------------------------
C
      WRITE(IOUT,5)
    5 FORMAT(1HO,'MAXIMUM HEAD CHANGE FOR EACH ITERATION:'/
     1    1HO,5(' HEAD CHANGE LAYER,ROW,COL')/1X,132('-'))
      WRITE (IOUT,10) (HDCG(J),(LRCH(I,J),I=1,3),J=1,KITER)
   10 FORMAT((1X,5(G12.4,' (',I3,',',I3,',',I3,')')))
      WRITE(IOUT,11)
   11 FORMAT(1HO)
C
      RETURN
C
      END
```

List of Variables for Module SSIP1P

| Variable | Range | Definition |
|---|---|---|
| HDCG | Package | DIMENSION (MXITER), Maximum head change for each iteration. |
| I | Module | Index for cell location. |
| IOUT | Global | Primary unit number for all printed output. IOUT = 6. |
| J | Module | Index for iterations. |
| KITER | Global | Iteration counter. Reset at the start of each time step. |
| LRCH | Package | DIMENSION (MXITER), Layer, row, and column of the cell containing the maximum head change (HDCG) for each iteration. |
| MXITER | Package | Maximum number of iterations. |

## Narrative for Module SSIP1I

Submodule SSIP1I calculates an iteration-parameter seed using model-conductance values and grid dimensions. Although a single seed is required, the method of calculation requires that three-directional seeds be calculated for each active cell. Then a cell seed, the minimum of the three, is selected. Finally, all the cell seeds are averaged to give the grid seed. This grid seed is then used to calculate the iteration parameters. The minimum cell seed is also printed.

Submodule SSIP1I performs its functions in the following order:

1. Calculate constants and initialize variables. In order to calculate the average cell seed, accumulators AVGSUM (sum of the cell seeds) and NODES (sum of the active cells for which a seed is calculated) are required. These are initialized to zero. WMINMN is used to store the smallest cell seed. Since this value must always be less than one, it is initialized to 1.0. The three coefficients, CCOL, CROW, and CLAY are set equal to $\pi^2/2(NCOL)^2$, $\pi^2/2(NROW)^2$, and $\pi^2/2(NLAY)^2$, respectively.

2. Loop through all cells, calculating a cell seed for each active cell.

   (a) Find the conductances from the cell to each of the six adjacent cells. Conductance across the grid boundary is set equal to zero.

   (b) Find the maximum and minimum of the two conductances in the row direction (DFMX, DFMN), in the column direction (BHMX, BHMN), and in the vertical direction (ZSMX, ZSMN). If the minimum is zero (which indicates that a neighbor is no flow), set the minimum equal to the maximum.

(c) Calculate three-directional seeds (WCOL, WROW, WLAY) using the relations

```
WCOL = CCOL/(1. + (BHMX + ZSMX)/DFMN);
WROW = CROW/(1. + (DFMX + ZSMX)/BHMN); and
WLAY = CLAY/(1. + (DFMX + BHMX)/ZSMN).
```

If the minimum conductance is zero (that is, both the minimum and the maximum are zero), set the seed equal to 1.0. This value will be ignored when the cell seed (the minimum-directional seed) is selected in step 2(d) because any valid seed will be less than 1.0.

(d) Select the minimum of the three-directional seeds as the cell seed. If it is the smallest cell seed used so far, store it in WMINMN. Accumulate the sum of the cell seeds and the total number of active cells so that the average of all cell seeds can be calculated in step 3.

3. Calculate the grid seed (the average cell seed) and print it along with the minimum seed.

4. Calculate and print iteration parameters using the grid seed with the relation

$$I_i = 1 - (SEED)^{\frac{i-1}{NPARM - 1}}$$

where

$I_i$ is the i-th iteration parameter, and

NPARM is the number of iteration parameters.

5. RETURN.

## Flow Chart for Module SSIP1I

Seed: the "grid seed" is the single parameter used to calculate the iteration parameters. To calculate the grid seed, several intermediate variables, called "cell seeds," are used. For each cell, three "directional seeds" are calculated. The minimum directional seed for a cell is the "cell seed." The "grid seed" is the average of the cell seeds.

AVGSUM is an accumulator to which each cell seed is added. It is then divided by the number of cells to obtain the average cell seed which is used as the grid seed.



```
      SUBROUTINE SSIP1I(CR,CC,CV,IBOUND,NPARM,W,NCOL,NROW,NLAY,
     1           IOUT)
C
C-----VERSION 1417 12MAY1987 SSIP1I
C     ******************************************************************
C     CALCULATE AN ITERATION PARAMETER SEED AND USE IT TO CALCULATE SIP
C     ITERATION PARAMETERS
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      DIMENSION CR(NCOL,NROW,NLAY),CC(NCOL,NROW,NLAY)
     1       ,CV(NCOL,NROW,NLAY),IBOUND(NCOL,NROW,NLAY),W(NPARM)
C
      DOUBLE PRECISION DWMIN,AVGSUM
C     ------------------------------------------------------------------
C
C1------CALCULATE CONSTANTS AND INITIALIZE VARIABLES
      PIEPIE=9.869604
      R=NROW
      C=NCOL
      ZL=NLAY
      CCOL=PIEPIE/(2.*C*C)
      CROW=PIEPIE/(2.*R*R)
      CLAY=PIEPIE/(2.*ZL*ZL)
      WMINMN=1.
      AVGSUM=0.
      NODES=0
C
C2------LOOP THROUGH ALL CELLS, CALCULATING A SEED FOR EACH CELL
C2------THAT IS ACTIVE
      DO 100 K=1,NLAY
      DO 100 I=1,NROW
      DO 100 J=1,NCOL
      IF(IBOUND(J,I,K).LE.0) GO TO 100
C
C2A-----CONDUCTANCE FROM THIS CELL
C2A-----TO EACH OF THE 6 ADJACENT CELLS
      D=0.
      IF(J.NE.1) D=CR(J-1,I,K)
      F=0.
      IF(J.NE.NCOL) F=CR(J,I,K)
      B=0.
      IF(I.NE.1) B=CC(J,I-1,K)
      H=0.
      IF(I.NE.NROW) H=CC(J,I,K)
      Z=0.
      IF(K.NE.1) Z=CV(J,I,K-1)
      S=0.
      IF(K.NE.NLAY) S=CV(J,I,K)
C
C2B-----FIND THE MAXIMUM AND MINIMUM OF THE 2 CONDUCTANCE COEFFICIENTS
C2B-----IN EACH PRINCIPAL COORDINATE DIRECTION
```

```
      DFMX=AMAX1(D,F)
      BHMX=AMAX1(B,H)
      ZSMX=AMAX1(Z,S)
      DFMN=AMIN1(D,F)
      BHMN=AMIN1(B,H)
      ZSMN=AMIN1(Z,S)
      IF(DFMN.EQ.0.) DFMN=DFMX
      IF(BHMN.EQ.0.) BHMN=BHMX
      IF(ZSMN.EQ.0.) ZSMN=ZSMX
C
C2C-----CALCULATE A SEED IN EACH PRINCIPAL COORDINATE DIRECTION
      WCOL=1.
      IF(DFMN.NE.0.) WCOL=CCOL/(1.+(BHMX+ZSMX)/DFMN)
      WROW=1.
      IF(BHMN.NE.0.) WROW=CROW/(1.+(DFMX+ZSMX)/BHMN)
      WLAY=1.
      IF(ZSMN.NE.0.) WLAY=CLAY/(1.+(DFMX+BHMX)/ZSMN)
C
C2D-----SELECT THE CELL SEED, WHICH IS THE MINIMUM SEED OF THE 3.
C2D-----SELECT THE MINIMUM SEED OVER THE WHOLE GRID.
      WMIN=AMIN1(WCOL,WROW,WLAY)
      WMINMN=AMIN1(WMINMN,WMIN)
C
C2E-----ADD THE CELL SEED TO THE ACCUMULATOR AVGSUM FOR USE
C2E-----IN GETTING THE AVERAGE SEED.
      DWMIN=WMIN
      AVGSUM=AVGSUM+DWMIN
      NODES=NODES+1
C
  100 CONTINUE
C
C3------CALCULATE THE AVERAGE SEED OF THE CELL SEEDS, AND PRINT
C3------THE AVERAGE AND MINIMUM SEEDS.
      TMP=NODES
      AVGMIN=AVGSUM
      AVGMIN=AVGMIN/TMP
      WRITE(IOUT,101) AVGMIN,WMINMN
  101 FORMAT(1H0,'AVERAGE SEED =',F11.8/1X,'MINIMUM SEED =',F11.8)
C
C4------CALCULATE AND PRINT ITERATION PARAMETERS FROM THE AVERAGE SEED
      P1=-1.
      P2=NPARM-1
      DO 50 I=1,NPARM
      P1=P1+1.
   50 W(I)=1.-AVGMIN**(P1/P2)
      WRITE(IOUT,150) NPARM,(W(J),J=1,NPARM)
  150 FORMAT(1H0,/,I5,' ITERATION PARAMETERS CALCULATED FROM',
     1       ' AVERAGE SEED:'//(10X,6E15.7))
C
C5------RETURN
      RETURN
      END
```

## List of Variables for Module SSIP1I

| Variable | Range | Definition |
|---|---|---|
| AVGMIN | Module | Mean WMIN. |
| AVGSUM | Module | Sum of all of WMIN's. |
| B | Module | Conductance between this node and the one to the rear. |
| BHMN | Module | Minimum of B and H (if the minimum is 0, it is the maximum). |
| BHMX | Module | Maximum of B and H. |
| C | Module | Number of columns. |
| CC | Global | DIMENSION (NCOL,NROW,NLAY), Conductance in the column direction. CC(J,I,K) contains conductance between nodes (J,I,K) and (J,I+1,K). |
| CCOL | Module | Intermediate factor. |
| CLAY | Module | Intermediate factor. |
| CR | Global | DIMENSION (NCOL,NROW,NLAY), Conductance in the row direction. CR(J,I,K) contains conductance between nodes (J,I,K) and (J+1,I,K). |
| CROW | Module | Intermediate factor. |
| CV | Global | DIMENSION (NCOL,NROW,NLAY-1), Conductance in the vertical direction. CV(J,I,K) contains conductance between nodes (J,I,K) and (J,I,K+1). |
| D | Module | Conductance between this node and the one to the left. |
| DFMN | Module | Minimum of D and F (if the minimum is 0, it is the maximum). |
| DFMX | Module | Maximum of D and F. |
| DWMIN | Module | Double precision WMIN. |
| F | Module | Conductance between this node and the one to the right. |
| H | Module | Conductance between this node and the one to the front. |
| I | Module | Index for rows. |
| IBOUND | Global | DIMENSION (NCOL,NROW,NLAY), Status of each cell.<br>< 0, constant-head cell<br>= 0, inactive cell<br>> 0, variable-head cell |

List of Variables for Module SSIP1I (Continued)

| Variable | Range | Definition |
|---|---|---|
| IOUT | Global | Primary unit number for all printed output. IOUT = 6. |
| J | Module | Index for columns. |
| K | Module | Index for layers. |
| NCOL | Global | Number of columns in the grid. |
| NLAY | Global | Number of layers in the grid. |
| NODES | Module | Number of variable-head (IBOUND > 0) cells in the grid. |
| NPARM | Package | Number of iteration parameters. |
| NROW | Global | Number of rows in the grid. |
| PIEPIE | Module | PI squared. |
| P1 | Module | Index for the number of parameters. |
| P2 | Module | NPARM-1. |
| R | Module | Number of rows. |
| S | Module | Conductance between this node and the one below. |
| TMP | Module | Temporary field for nodes. |
| W | Package | DIMENSION (NPARM), Iteration parameters. |
| WCOL | Module | Seed in the column direction for a cell. |
| WLAY | Module | Seed in the layer direction for a cell. |
| WMIN | Module | Minimum of (WCOL, WLAY, WROW). |
| WMINMN | Module | Minimum WMIN. |
| WROW | Module | Seed in the row direction for a cell. |
| Z | Module | Conductance between this node and the one above. |
| ZL | Module | Number of layers. |
| ZSMN | Module | Minimum of Z and S (if minimum is 0, it is the maximum). |
| ZSMX | Module | Maximum of Z and S. |